Polynômes

III) Développements limités

III-1 Un exemple

1.

```
> f:=x->ln(1+x)/(x+2);
```

$$f := x \to \frac{\ln(x+1)}{x+2}$$

2.

```
> DL:=series(f(x),x=0,4);
```

$$DL := \frac{1}{2}x - \frac{1}{2}x^2 + \frac{5}{12}x^3 + O(x^4)$$

3.(a)

```
> P:=convert(DL,polynom);
```

$$P := \frac{1}{2}x - \frac{1}{2}x^2 + \frac{5}{12}x^3$$

```
> plot([f(x),P(x)],x=-5..5,y=-5..5,color=[black,blue]);
```



3.(b)

```
> coeff(P,x,3);
```

$$\frac{5}{12}$$

III-2 Des exercices

1.

```
> restart;
  DL:=series(sinh(x)-(a*x+b*x^3+c*x^5)/(1+d*x^2+e*x^4),x=0,10);
  P:=convert(DL,polynom):
  C1:=coeff(P,x,1):
  C3:=coeff(P,x,3):
  C5:=coeff(P,x,5):
  C7:=coeff(P,x,7):
  C9:=coeff(P,x,9):
  sol:=solve({C1=0,C3=0,C5=0,C7=0,C9=0},{a,b,c,d,e});
```

$$DL := (1-a)\,x + \left(\frac{1}{6} - b + a\,d\right) x^3 + \left(-c + a\,e - (-b + a\,d)\,d + \frac{1}{120}\right) x^5 + \left(-(-b + a\,d)\,e - (-c + a\,e + d\,b - a\,d^2)\,d + \frac{1}{5040}\right) x^7 + \left(\frac{1}{362880} - (-c + a\,e + d\,b - a\,d^2)\,e - (e\,b - 2\,e\,a\,d + d\,c - d^2\,b + a\,d^3)\,d\right) x^9 + \mathrm{O}(x^{10})$$

$$sol := \{a = 1,\ e = \frac{5}{11088},\ d = \frac{-13}{396},\ b = \frac{53}{396},\ c = \frac{551}{166320}\}$$

La solution est unique. Nous avons donc trouvé la solution optimale.

```
> assign(sol);
  plot([sinh(x),(a*x+b*x^3+c*x^5)/(1+d*x^2+e*x^4)],x=-7..7,color=[
  black,blue]);
```



2.(a)

```
> restart;
  DL:=series(a/sin(x)+b/tan(x)+c*x/cos(x)+d/x,x=0,3);
```

```
P:=convert(DL,polynom);
```

$$DL := (a+b+d)\,x^{-1} + O(x)$$

$$P := \frac{a+b+d}{x}$$

La condition nécessaire et suffisante pour f se prolonge par continuité en 0 est donc :

```
> coeff(P,x,-1)=0;
```

$$a+b+d=0$$

2.(b)

```
> DL:=series(a/sin(x)+b/tan(x)+c*x/cos(x)+d/x,x=0,7);
  P:=convert(DL,polynom);
```

$DL :=$

$$(a+b+d)\,x^{-1} + \left(-\frac{b}{3}+c+\frac{a}{6}\right)x + \left(\frac{7\,a}{360}-\frac{b}{45}+\frac{c}{2}\right)x^3 + \left(\frac{31\,a}{15120}+\frac{5\,c}{24}-\frac{2\,b}{945}\right)x^5 + O(x^7)$$

$$P := \frac{a+b+d}{x} + \left(-\frac{b}{3}+c+\frac{a}{6}\right)x + \left(\frac{7\,a}{360}-\frac{b}{45}+\frac{c}{2}\right)x^3 + \left(\frac{31\,a}{15120}+\frac{5\,c}{24}-\frac{2\,b}{945}\right)x^5$$

```
> Cm1:=coeff(P,x,-1);
  C1:=coeff(P,x,1);
  C3:=coeff(P,x,3);
  C5:=coeff(P,x,5);
```

$$Cm1 := a+b+d$$

$$C1 := -\frac{b}{3}+c+\frac{a}{6}$$

$$C3 := \frac{7\,a}{360}-\frac{b}{45}+\frac{c}{2}$$

$$C5 := \frac{31\,a}{15120}+\frac{5\,c}{24}-\frac{2\,b}{945}$$

```
> solve({Cm1=0,C1=0,C3=0,C5=0},{a,b,c,d});
```

$$\{a=0,\ d=0,\ c=0,\ b=0\}$$

(a,b,c,d) est non nul. Nous devons donc enlevé la quatrième équation.

```
> solve({Cm1=0,C1=0,C3=0},{a,b,c,d});
```

$$\{b=-23\,c,\ d=75\,c,\ a=-52\,c,\ c=c\}$$

Il y a donc une droite solution au problème.

Remarque : Le problème était un problème linéaire homogène. Les solutions du problèmes forment donc un sous-espace vectoriel de R^4.

IV) Détermination de coefficients

1.

```
> restart;
  P:=X->X^6+sqrt(3)*X^4+a*X^2+b*X+c;
  solve({P(1)=0,P(2)=0,P(3)=0},{a,b,c});
```

$$P := X \to X^6 + \sqrt{3}\,X^4 + a\,X^2 + b\,X + c$$
$$\{c = -540 - 36\sqrt{3},\ b = 840 + 60\sqrt{3},\ a = -301 - 25\sqrt{3}\}$$

2.(a)

Puisque a et b sont réels, le polynôme P est un polynôme réel et donc 1+2i est une racine de P si, et seulement si, 1-2i est une racine de P.

```
> restart;
  P:=X->X^4+a*X^3+sqrt(3)*X+b;
  sol:=solve({P(1+2*I)=0,P(1-2*I)=0},{a,b});
```

$$P := X \to X^4 + a\,X^3 + \sqrt{3}\,X + b$$
$$sol := \{a = -12 + \sqrt{3},\ b = -125 + 10\sqrt{3}\}$$

2.(b)

```
> assign(sol);
  factor(evalf(P(X)),complex);
  factor(P(X));
```

$$(X + 2.080978690)\,(X - 1.000000000 + 2.000000000\,I)\,(X - 1.000000000 - 2.000000000\,I)\,(X - 10.34892788)$$
$$(X^2 - 2\,X + 5)\,(X^2 + \sqrt{3}\,X - 10\,X - 25 + 2\sqrt{3})$$

Nous n'obtenons pas la factorisation dans R en facteurs irréductibles.
Nous pouvons améliorer nos réponses :

```
> solve(P(X),X);
```

$$1 + 2\,I,\ 1 - 2\,I,\ -\frac{\sqrt{3}}{2} + 5 + \frac{\sqrt{203 - 28\sqrt{3}}}{2},\ -\frac{\sqrt{3}}{2} + 5 - \frac{\sqrt{203 - 28\sqrt{3}}}{2}$$

```
> factor(P(X),I);
```

$$(X^2 + \sqrt{3}\,X - 10\,X - 25 + 2\sqrt{3})\,(X - 1 - 2\,I)\,(X - 1 + 2\,I)$$

```
> factor(P(X),{I,sqrt(203-28*sqrt(3))});
```

$$-(-X + 1 + 2\,I)\,(X - 1 + 2\,I)\,(-10 + \sqrt{3} - \sqrt{203 - 28\sqrt{3}} + 2\,X)\,(-10 + \sqrt{3} + \sqrt{203 - 28\sqrt{3}} + 2\,X)\,/\,4$$

Remarque : Pour factoriser, Maple utilise les nombres formés par sommes, produits et fractions des rationnels et des coefficients du polynôme. On peut lui spécifier d'autres nombres pour améliorer la factorisation comme ci-dessus.

V) Racines d'un polynôme

1.

```
> restart;
  P1:=X^4-15*X^2-10*X+24;
  P2:=X^4-4*X^3-3*X^2+14*X-8;
  P3:=X^4+3*X^2-6*X+10;
  P4:=X^4-X^3-3*X^2+X+1;
  L1:=[solve(P1,X)];
  L2:=[solve(P2,X)];
```

```
L3:=[solve(P3,X)];
L4:=map(allvalues,[solve(P4,X)]);
```

$$P1 := X^4 - 15\,X^2 - 10\,X + 24$$

$$P2 := X^4 - 4\,X^3 - 3\,X^2 + 14\,X - 8$$

$$P3 := X^4 + 3\,X^2 - 6\,X + 10$$

$$P4 := X^4 - X^3 - 3\,X^2 + X + 1$$

$$L1 := [1, -2, -3, 4]$$

$$L2 := [-2, 4, 1, 1]$$

$$L3 := [1 + I, 1 - I, -1 + 2\,I, -1 - 2\,I]$$

$$L4 := \left[\frac{1}{4} - \frac{\sqrt{5}}{4} + \frac{\sqrt{22 - 2\sqrt{5}}}{4}, \frac{1}{4} + \frac{\sqrt{5}}{4} + \frac{\sqrt{22 + 2\sqrt{5}}}{4}, \frac{1}{4} + \frac{\sqrt{5}}{4} - \frac{\sqrt{22 + 2\sqrt{5}}}{4}, \frac{1}{4} - \frac{\sqrt{5}}{4} - \frac{\sqrt{22 - 2\sqrt{5}}}{4}\right]$$

2.(a)

```
> FactC:=proc(L,X)
         local k,P;
  P:=1;
  for k from 1 to nops (L) do
     P:=P*(X-L[k]);
  od;
  P;
  end;
```

$FactC :=$

$\mathbf{proc}(L, X)\ \mathbf{local}\ k, P;\ P := 1;\ \mathbf{for}\ k\ \mathbf{to}\ \mathrm{nops}(L)\ \mathbf{do}\ P := P*(X - L[k])\ \mathbf{end\ do};\ P\ \mathbf{end\ proc}$

2.(b)

```
> P1=FactC(L1,X);
  P2=FactC(L2,X);
  P3=FactC(L3,X);
  P4=FactC(L4,X);
```

$$X^4 - 15\,X^2 - 10\,X + 24 = (X - 1)\,(X + 3)\,(X + 2)\,(X - 4)$$

$$X^4 - 4\,X^3 - 3\,X^2 + 14\,X - 8 = (X + 2)\,(X - 4)\,(X - 1)^2$$

$$X^4 + 3\,X^2 - 6\,X + 10 = (X - 1 - I)\,(X - 1 + I)\,(X + 1 - 2\,I)\,(X + 1 + 2\,I)$$

$$X^4 - X^3 - 3\,X^2 + X + 1 = \left(X - \frac{1}{4} + \frac{\sqrt{5}}{4} - \frac{\sqrt{22 - 2\sqrt{5}}}{4}\right)\left(X - \frac{1}{4} - \frac{\sqrt{5}}{4} - \frac{\sqrt{22 + 2\sqrt{5}}}{4}\right)\left(X - \frac{1}{4} - \frac{\sqrt{5}}{4} + \frac{\sqrt{22 + 2\sqrt{5}}}{4}\right)\left(X - \frac{1}{4} + \frac{\sqrt{5}}{4} + \frac{\sqrt{22 - 2\sqrt{5}}}{4}\right)$$

3.(a)

```
> FactR:=proc(L,X)
         local k,P;
  P:=1;
```

```
for k from 1 to nops(L) do
   if type(L[k],realcons) then
      P:=P*(X-L[k]);
   else
      P:=P*expand((X-L[k])*(X-L[k+1]));
      k:=k+1; # Il est nécessaire de passer le terme suivant de
la liste.
   fi;
od;
P;
end;
```

$FactR$:= **proc**(L, X)

local k, P;

$P := 1$;

for k **to** nops(L) **do**

if type($L[k]$, $realcons$) **then** $P := P*(X - L[k])$

else $P := P*\mathrm{expand}((X - L[k])*(X - L[k+1]))$; $k := k + 1$

end if

end do;

P

end proc

3.(b)

Remarquons que les listes L1, L2, L3 et L4 vérifient l'hypothèse demandée.

```
> P1=FactR(L1,X);
  P2=FactR(L2,X);
  P3=FactR(L3,X);
  P4=FactR(L4,X);
```

$$X^4 - 15\,X^2 - 10\,X + 24 = (X-1)\,(X+3)\,(X+2)\,(X-4)$$

$$X^4 - 4\,X^3 - 3\,X^2 + 14\,X - 8 = (X+2)\,(X-4)\,(X-1)^2$$

$$X^4 + 3\,X^2 - 6\,X + 10 = (X^2 - 2\,X + 2)\,(X^2 + 2\,X + 5)$$

$$X^4 - X^3 - 3\,X^2 + X + 1 = \left(X - \frac{1}{4} + \frac{\sqrt{5}}{4} - \frac{\sqrt{22 - 2\sqrt{5}}}{4}\right)\left(X - \frac{1}{4} - \frac{\sqrt{5}}{4} - \frac{\sqrt{22 + 2\sqrt{5}}}{4}\right)\left(X - \frac{1}{4} - \frac{\sqrt{5}}{4} + \frac{\sqrt{22 + 2\sqrt{5}}}{4}\right)\left(X - \frac{1}{4} + \frac{\sqrt{5}}{4} + \frac{\sqrt{22 - 2\sqrt{5}}}{4}\right)$$

4.(a)

```
> RelCoefRac:=proc(P,L,X)
              local b0,b1,b2,b3;
  b0:=evalb(expand(coeff(P,X,0)/coeff(P,X,4))=expand(L[1]*L[2]*L[3
  ]*L[4]));
  b1:=evalb(expand(coeff(P,X,1)/coeff(P,X,4))=-expand(L[1]*L[2]*L[
  3]+L[1]*L[2]*L[4]+L[1]*L[3]*L[4]+L[2]*L[3]*L[4]));
  b2:=evalb(expand(coeff(P,X,2)/coeff(P,X,4))=expand(L[1]*L[2]+L[1
```

```
]*L[3]+L[1]*L[4]+L[2]*L[3]+L[2]*L[4]+L[3]*L[4]));
b3:=evalb(expand(coeff(P,X,3)/coeff(P,X,4))=-expand(L[1]+L[2]+L[
3]+L[4]));
b0 and b1 and b2 and b3;
end;
```

RelCoefRac := **proc**(*P*, *L*, *X*)

local *b0*, *b1*, *b2*, *b3*;

$b0$:= evalb(expand(coeff(P, X, 0) / coeff(P, X, 4)) = expand($L[1]*L[2]*L[3]*L[4]$));

$b1$:= evalb(expand(coeff(P, X, 1) / coeff(P, X, 4)) = −expand($L[1]*L[2]*L[3] + L[1]*L[2]*L[4] + L[1]*L[3]*L[4] + L[2]*L[3]*L[4]$));

$b2$:= evalb(expand(coeff(P, X, 2) / coeff(P, X, 4)) = expand($L[1]*L[2] + L[1]*L[3] + L[1]*L[4] + L[2]*L[3] + L[2]*L[4] + L[3]*L[4]$));

$b3$:= evalb(expand(coeff(P, X, 3) / coeff(P, X, 4)) = −expand($L[1] + L[2] + L[3] + L[4]$));

b0 **and** *b1* **and** *b2* **and** *b3*

end proc

```
> RelCoefRac(P1,L1,X);
  RelCoefRac(P2,L2,X);
  RelCoefRac(P3,L3,X);
  RelCoefRac(P4,L4,X);
```

true

true

true

true

```
> RelCoefRac(P1,L2,X);
```

false

```
>
```